The Whiley Language Specification

David J. Pearce School of Engineering and Computer Science Victoria University of Wellington, New Zealand djp@ecs.vuw.ac.nz

February 10, 2014

Contents

1	Intr	oduction 4													
	1.1	Background													
	1.2	Goals													
	1.3	History													
2	Lovi	cal Structure 6													
4	2.1 Line Terminators														
	2.1														
	2.2														
		Comments													
	2.4														
	2.5	Keywords													
	2.6	Literals													
		2.6.1 Null Literal													
		2.6.2 Boolean Literals													
		2.6.3 Byte Literals													
		2.6.4 Integer Literals													
		2.6.5 Real Literals													
		2.6.6 Character Literals													
		2.6.7 String Literals													
3	Som	rce Files 10													
	3.1	Compilation Units													
	3.2	Packages													
	3.3	Names													
	3.4	Imports													
	3.5	Named Declarations													
	3.3	3.5.1 Access Control													
		V 1													
		3.5.3 Constant Declarations													
		3.5.4 Function Declarations													
		3.5.5 Method Declarations													
4	Types														
	4.1	Overview													
		4.1.1 Type Descriptors													
		4.1.2 Type Patterns													
	4.2	Primitive Types													
		4.2.1 Null													
		4.2.2 Booleans													
		4.2.3 Bytes													
		4.2.4 Integers													
		4.2.5 Partianals 19													

		4.2.6	Character	·s					 		 				 		19
		4.2.7	Any						 		 						19
		4.2.8	Void						 		 		 		 		20
	4.3	Tuples							 		 						20
	4.4																
	4.5		ces														
	4.6		ls														22
	4.7		ons														23
	1.,		Sets														23
			Maps														23
			Lists														
	4.8																
			ns and Me														24 25
	4.9																
			tions														25
			ns														26
			ve Types														26
	4.13		e Types														27
			Effective														27
		4.13.2	Effective	Record	ds .				 		 				 		27
		4.13.3	Effective	Collec	tion	s.			 		 				 		27
	4.14	Semanti	ics						 		 						27
			Equivalen														28
			Subtyping														
			71 0														
5	State	ements															30
	5.1	Blocks							 		 		 				30
	5.2		Statement														
	5.3		nent State														
	5.4		Statemen														
	5.5		tatement .														
	5.6		e Stateme														
	5.7		Statement														
	5.8		le Statem														
	5.9		ement .														
			ment														
			tatement														
			Statement														
			atement .														34
			Statement														
	5.15	Throw S	Statement						 		 				 		34
	5.16	Try/Cat	ch Statem	ent .					 		 						34
	5.17	Variable	Declarat	ion Sta	item	ent			 		 						35
6	Expr	ressions															36
	6.1	Tuple E	xpression	s					 		 		 		 		36
	6.2		pressions														36
	6.3		Expression														36
	6.4	_	Expression														37
	6.5		on Expres														37
	6.6		ier Expres														38
	6.7	_	Expression														38
	6.8		Expression														38
		_															
	6.9		pressions Multipli										•	 •	 •	 •	38
	h 10	A ddifive	2/ Winifinia	cative	HVN	reco	2101	C									39

Glossary								
6.13	Dereference Expressions	9						
6.12	Term Expressions	9						
6.11	Access Expressions	9						

Chapter 1

Introduction

This document provides a specification of the *Whiley Programming Language*. Whiley is a hybrid imperative and functional programming language designed to produce programs with as few errors as possible. Whiley allows explicit specifications to be given for functions, methods and data structures, and employs a *verifying compiler* to check whether programs meet their specifications. As such, Whiley is ideally suited for use in *safety critical systems*. However, there are many benefits to be gained from using Whiley in a general setting (e.g. improved documentation, maintainability, reliability, etc). Finally, this document is *not* intended as a general introduction to the language, and the reader is referred to alternative documents for learning the language^[1].

1.1 Background

Reliability of large software systems is a difficult problem facing software engineering, where subtle errors can have disastrous consequences. Infamous examples include: the Therac-25 disaster where a computer-operated X-ray machine gave lethal doses to patients^[2]; the 1988 worm which reeked havoc on the internet by exploiting a buffer overrun^[3]; the 1991 Patriot missile failure where a rounding error resulted in the missile catastrophically hitting a barracks^[4]; and, the Ariane 5 rocket which exploded shortly after launch because of an integer overflow, costing the ESA an estimated \$500 million^[5].

Around 2003, Hoare proposed the creation of a *verifying compiler* as a grand challenge for computer science ^[6]. A verifying compiler "*uses automated mathematical and logical reasoning to check the correctness of the programs that it compiles.*" There have been numerous attempts to construct a verifying compiler system, although none has yet made it into the mainstream. Early examples include that of King ^[7], Deutsch ^[8], the Gypsy Verification Environment ^[9] and the Stanford Pascal Verifier ^[10]. More recently, the Extended Static Checker for Modula-3 ^[11] which became the Extended Static Checker for Java (ESC/Java) — a widely acclaimed and influential work ^[12]. Building on this success was JML and its associated tooling which provided a standard notation for specifying functions in Java ^[13]. Finally, Microsoft developed the Spec# system which is built on top of C#^[14].

1.2 Goals

The Whiley Programming Language has been designed from scratch in conjunction with a verifying compiler. The intention is to provide an open framework for research in automated software verification. The initial goal is to automatically eliminate common errors, such as *null dereferences*, *array-out-of-bounds*, *divide-by-zero* and more. In the future, the intention is to consider more complex issues, such as termination, proof-carrying code and user-supplied proofs.

1.3 History

Development of the Whiley programming language begun in 2009 by Dr. David J. Pearce, at the time a lecturer in Computer Science at Victoria University of Wellington. The accompanying website http://whiley.org went live in 2010, making the first versions of Whiley available for download. Since then, Whiley has been in constant development with the majority of contributions being made by the original author. Several scientific papers have published on different aspects of the language, including:

- Implementing a Language with Flow-Sensitive and Structural Typing on the JVM. David J. Pearce and James Noble. In *Proceedings of the Workshop on Bytecode Semantics, Verification, Analysis and Transformation (BYTECODE)*, 2011.
- Sound and Complete Flow Typing with Unions, Intersections and Negations, David J. Pearce. In Proceedings of the Conference on Verification, Model Checking and Abstract Interpretation (VMCAI), pages 335–354, 2013
- A Calculus for Constraint-Based Flow Typing. David J. Pearce. In *Proceedings of the Workshop on Formal Techniques for Java-like Languages (FTFJP)*, Article 7, 2013.
- Whiley: a Platform for Research in Software Verification. David J. Pearce and Lindsay Groves. In *Proceedings of the Conference on Software Language Engineering (SLE)*, pages 238-248, 2013
- Reflections on Verifying Software with Whiley. David J. Pearce and Lindsay Groves. In *Proceedings of the Workshop on Formal Techniques for Safety-Critical Software (FTSCS)*, 2013

Chapter 2

Lexical Structure

This chapter specifies the lexical structure of the Whiley programming language. Programs in Whiley are organised into one or more *source files* written in Unicode. The Whiley language uses *indentation syntax* to delimit blocks and statements, rather than curly-braces (or similar) as found in many other languages.

2.1 Line Terminators

A Whiley compiler splits the sequence of (Unicode) input characters into lines by identifying *line terminators*:

2.2 Indentation

After splitting the input characters into lines, a Whiley compiler then identifies the *indentation* of each line. This is necessary because Whiley employs indentation syntax meaning that indentation is significant in the meaning of Whiley programs.

Here, $\hat{}$ demarcates the start of a line and, hence, indentation may only occur at the beginning of a line. Indentation may be compared using the \leq comparator, such that $i \leq ir$ always holds (where i is some indentation and r is either empty or represents additional indentation). In other words, some indentation i is considered less-than-or-equal to another piece of indentation ir which includes the first as a prefix. This comparator is important for delimiting $statement\ blocks$ (§5.1).

2.3 Comments

There are two kinds of comments in Whiley: line comments and block comments:



2.4 Identifiers

An identifier is a sequence of one or more letters or digits which starts with a letter.

```
Ident ::= Letter(Letter|Digit)*
Letter ::= -|a|...|z|A|...|Z
Digit ::= 0|1|2|3|4|5|6|7|8|9
```

Letters include lowercase and uppercase alphabetic characters (i.e. a-z and A-Z) and the underscore (_).

2.5 Keywords

The following strings are reserved for use as keywords and may not be used as identifiers:



The following strings are reserved for use as *keywords*, but may additionally be used as identifiers in certain contexts:

```
KeywordIdentifier ::= constant | from | type
```

2.6 Literals

A *literal* is a source-level entity which describes a value of primitive type (§4.2).

```
Literal ::= NullLiteral
| BoolLiteral
| ByteLiteral
| IntLiteral
| RealLiteral
| CharLiteral
| StringLiteral
```

2.6.1 Null Literal

The null type (§4.2.1) has a single value expressed as the null literal.

```
NullLiteral ::= null
```

2.6.2 Boolean Literals

The bool type (§4.2.2) has two values expressed as the true and false literals.

2.6.3 Byte Literals

The **byte** type (§4.2.3) has 256 values which are expressed as sequences of binary digits, followed by the suffix "b" (e.g. 0101b).

ByteLiteral
$$:= (0|1)^+$$
 b

Byte literals do not need to contain exactly eight digits and, when fewer digits are given, are padded out to eight digits by appending zero's from the left (e.g. 00101b becomes 00000101b).

2.6.4 Integer Literals

The int type (§4.2.4) represents the infinite set of integer values which are expressed as sequences of numeric or hexadecimal digits (e.g. 123456, 0xffaf, etc).

IntLiteral
$$::= (0 | \dots | 9)^+$$

$$| 0 | x | (0 | \dots | 9 | a | \dots | f | A | \dots | F)^+$$

Since integer values in Whiley are of arbitrary size ($\S4.2.4$), there is no limit on the size of an integer literal.

2.6.5 Real Literals

The **real** type (§4.2.5) represents the infinite set of rational values which are expressed as sequences of numeric digits separated by a period (e.g. 1.0, 0.223, 12.55, etc).

RealLiteral
$$::= (0 | \dots | 9)^+ \cdot (0 | \dots | 9)^+$$

2.6.6 Character Literals

A *character literal* is expressed as a single character or an escape sequence enclosed in single quotes (e.g. 'c').

2.6.7 String Literals

A *string literal* is expressed as a sequence of zero or more characters or escape sequences enclosed in double quotes (e.g. "Hello_World").

Chapter 3

Source Files

Whiley programs are split across one or more source files which are compiled into *WyIL files* prior to execution. Source files contain declarations which describe the functions, methods, data types and constants which form the program. Source files are grouped together into coherent units called *packages*.

3.1 Compilation Units

Two kinds of *compilation unit* are taken into consideration when compiling a Whiley source file: other source files; and, binary WyIL files. The Whiley Intermediate Language (WyIL) file format is described elsewhere, but defines a binary representation of a Whiley source file. When one or more Whiley source files are compiled together, a *compilation group* is formed. External symbols encountered during compilation are first resolved from the compilation group, and then from previously WyIL files.

3.2 Packages

Programs in Whiley are organised into packages to help reduce name conflicts and provide some grouping of related concepts. A Whiley source file may provide an optional package declaration to identify the package it belongs to. This declaration must occur at the beginning of the source file.

```
PackageDecl ::= package Ident (. Ident)*
```

Any source file which does not provide a package declaration is considered to be in the *default* package.

3.3 Names

There are four functional entities which can be defined within a Whiley source file: *type declarations* (§3.5.2), *constant declarations* (§3.5.3), *function declarations* (§3.5.4) and *method declarations* (§3.5.5). These define *named entities* which may be referenced from other compilation units. Every named entity has a unique *fully-qualified* name constructed from the enclosing package name, the source file name and the declared name. For example:

Graphics.whiley

```
package tracer

type Point is { int x, int y }

constant Origin is { x: 0, y: 0 }
```

This declares two named entities: tracer.Graphics.Point and tracer.Graphics.Origin.

Two named entities may *clash* if they have the same fully qualified name and are in the same category. There are three entity categories: *types*, *constants* and *functions/methods*. The following illustrates a common pattern:

```
type Point is { int x, int y }
function Point(int x, int y) => Point:
    return {x: x, y: y}
```

Here, two named entities share the same fully qualified named. This is permitted because they are in different categories.

3.4 Imports

When performing *name resolution*, a Whiley compiler first attempts to resolve names within the same source file. For any remaining unresolved, the compiler examines imported entities in reverse declaration order. Entities are imported using an import declaration:

A declaration of the form "import some.pkg.File" imports the compilation unit "File" residing in package "some.pkg". Named entities (e.g. "Entity") within that compilation unit can then be referenced using a partially qualified name which omits the package component (e.g. "File.Entity"). A declaration of the form "import Entity from some.pkg.File" imports the named entity "Entity" from the compilation unit "File" residing in package "some.pkg". Note, this does not import the compilation unit "some.pkg.File" (and, hence, does not subsume the statement "import some.pkg.File").

A wildcard may be used in place of the compilation unit name (e.g. "import some.pkg.*") to import all compilation units within the given package. A package match may be used in place of some of all of the package component (e.g. "import some..File") to import all compilation units with matching name and package prefix and/or suffix. A wildcard may be used in place of the entity name (e.g. "import * from some.pkg.File") to import all named entities within the given compilation unit.

3.5 Named Declarations

Camel case

3.5.1 Access Control

3.5.2 Type Declarations

A type declaration declares a named type within a Whiley source file. The declaration may refer to named types in this or other source files and may also recursively refer to itself (either directly or indirectly).

```
TypeDecl ::= type Ident is TypePattern [where Expr]
```

The optional **where** clause defines a *boolean expression* which holds for any instance of this type. This is often referred to as the type *invariant* or *constraint*. Variables declared within the *type pattern* may be referred to within the optional **where** clause.

Examples. Some simple examples illustrating type declarations are:

```
// Define a simple point type
type Point is { int x, int y }

// Define the type of natural numbers
type nat is (int x) where x >= 0
```

The first declaration defines an unconstrained record type named Point, whilst the second defines a constrained integer type nat.

Notes. A convention is that type declarations for *records* or *unions of records* begin with an upper case character (e.g. Point above). All other type declarations begin with lower case. This reflects the fact that records are most commonly used to describe objects in the domain.

3.5.3 Constant Declarations

A *constant declaration* declares a named constant within a Whiley source file. The declaration may refer to named constants in this or other source files, although it may not refer to itself (either directly or indirectly).

```
ConstantDecl ::= constant Ident is Expr
```

The given *constant expression* is evaluated at *compile time* and must produce a constant value. This prohibits the use of function or method calls within the constant expression. However, general operators (e.g., for arithmetic) are permitted.

Examples. Some simple examples to illustrate constant declarations are:

```
// Define the well-known mathematical constant to 10 decimal places.

constant PI is 3.141592654

// Define a constant expression which is twice PI

constant TWO_PI is PI * 2.0
```

The first declaration defines the constant PI to have the **real** value 3.141592654. The second declaration illustrates a more interesting constant expression which is evaluated to 6.283185308 at compile time.

Notes. A convention is that constants are named in upper case with underscores separating words (i.e. as in TWO PI above).

3.5.4 Function Declarations

A function declaration defines a function within a Whiley source file. Functions are pure and may not have side-effects. This means they are guaranteed to return the same result given the same arguments, and are permitted within specifications (i.e. in type invariants, loop invariants, and function/method preconditions or postconditions). Functions may call other functions, but may not call other methods. Functions may not allocate memory on the heap and/or instigate concurrent computation.

```
FunctionDecl ::= function Ident TypePattern => TypePattern (
throws Type | requires Expr | ensures Expr
)* : Block
```

The first type pattern (i.e. before "=>") is referred to as the *parameter*, whilst the second is referred to as the *return*. There are three kinds of optional clause which follow:

- Throws clause. This defines the exceptions which may be thrown by this function. Multiple clauses may be given, and these are taken together as a union. Furthermore, the convention is to specify throws clause(s) first.
- Requires clause(s). These define constraints on the permissible values of the parameters on entry to the function, and are often collectively referred to as the precondition. These expressions may refer to any variables declared within the parameter type pattern. Multiple clauses may be given, and these are taken together as a conjunction. Furthermore, the convention is to specify the requires clause(s) before any ensures clause(s).
- Ensures clause(s). These define constraints on the permissible values of the function's return value, and are often collectively referred to as the postcondition. These expressions may refer to any variables declared within either the parameter or return type pattern. Multiple clauses may be given, and these are taken together as a conjunction. Furthermore, the convention is to specify ensures clause(s) last.

Examples. The following function declaration provides a small example to illustrate:

```
function max(int x, int y) => (int z)
// return must be greater than either parameter
ensures x <= z && y <= z
// return must equal one of the parmaeters
ensures x == z | | y == z:
    // implementation
    if x > y:
        return x
else:
        return y
```

This defines the specification and implementation of the well-known \max () function which returns the largest of its parameters. This does not throw any exceptions, and does not enforce any preconditions on its parameters.

3.5.5 Method Declarations

A *method declaration* defines a method within a Whiley source file. Methods are *impure* and may have side-effects. Thus, they cannot be used within specifications (i.e. in type invariants, loop invariants, and function/method preconditions or postconditions). However, unlike functions, they methods call other functions and/or methods (including native methods). They may also allocate memory on the heap, and/or instigate concurrent computation.

```
MethodDecl ::= method Ident TypePattern => TypePattern (
throws Type | requires Expr | ensures Expr
)* : Block
```

The first type pattern (i.e. before "=>") is referred to as the *parameter*, whilst the second is referred to as the *return*. The three optional clauses are defined identically as for function declarations (§3.5.4).

Examples. The following method declaration provides a small example to illustrate:

```
// Define the well-known concept of a linked list
type LinkedList is null | { &LinkedList next, int data }

// Define a method which inserts a new item onto the end of the list
method insertAfter(&LinkedList list, int item):
    if *list is null:
        // reached the end of the list, so allocate new node
        *list = new { next: null, data: item }

else:
        // continue traversing the list
        insertAfter(list→next, item)
```

Chapter 4

Types

The Whiley programming language is *statically typed*, meaning that every expression has a type determined at compile time. Furthermore, evaluating an expression is guaranteed to yield a value of its type. Whiley's *type system* governs how the type of any variable or expression is determined. Whiley's type system is unusual in that it incorporates *union types* (§4.9), *intersection types* (§4.10) and *negation types* (§4.11), as well as employing *flow typing* and *structural typing*.

4.1 Overview

Types in Whiley are unusual (in part) because there is a large gap between their *syntactic* description and their underlying *semantic* meaning. In most programming languages (e.g. Java), this gap is either small or non-existent and, hence, there is little to worry about. However, in Whiley, we must tread carefully to avoid confusion. The following example attempts to illustrate this gap between the syntax and semantics of types:

```
int|null id(null|int x):
    return x
```

In this function we see two distinct *type descriptors* expressed in the program text, namely "int |null" and "null|int". Type descriptors occur at the source-level and describe *types* which occur at the abstract (or underlying) level. In this particular case, we have two distinct type descriptors which describe the *same* underlying type. We will often refer to types as providing the semantic (i.e. meaning) of type descriptors.

4.1.1 Type Descriptors

Type descriptors provide syntax for describing types and, in the remaining sections of this chapter, we explore the range of types supported in Whiley. The top-level grammar for type descriptors is:

4.1.2 Type Patterns

Type patterns associate variables with types and their subcomponents and can be used to declare variables and/or *destructuring* types into variables. Type patterns are a source-level entity which are similar to type descriptors. The top-level grammar for type patterns is:

Type patterns do not exist for all compound structures — only those where a value is guaranteed to exist which could be associated with a variable.

4.2 Primitive Types

Primitive types are the atomic building blocks of all types in Whiley.

```
PrimitiveType ::=

AnyType
VoidType
NullType
BoolType
ByteType
CharType
IntType
RealType
```

4.2.1 Null

The null type is typically used to show the absence of something. It is distinct from void, since variables can hold the special **null** value (where as there is no special "**void**" value). The set of values defined by the type **null** is the singleton set containing exactly the **null** value. Variables of **null** type support only equality (§6.5) and inequality comparisons (§6.5). The **null** value is particularly useful for representing optional values and terminating recursive types.

```
NullType ::= null
```

The following illustrates a simple example of the **null** type:

```
type Tree is null | { int data, Tree left, Tree right }
function height(Tree t) => int:
    if t is null:
        // height of empty tree is zero
        return 0
else:
        // height is this node plus maximum height of subtrees
        return 1 + Math.max(height(t.left), height(t.right))
```

This defines Tree — a recursive type — which is either empty (i.e. null) or consists of a field data and two subtrees, left and right. The height function calculates the height of a Tree as the longest path from the root through the tree.

Notes. With all of the problems surrounding **null** and NullPointerExceptions in languages like Java and C, it may seem that this type should be avoided. However, it remains a very useful abstraction around (e.g. for terminating recursive types) and, in Whiley, is treated in a completely safe manner (unlike e.g. Java).

4.2.2 Booleans

The **bool** type represents the set of boolean values (i.e. true and false). Variables of **bool** type support equality ($\S6.5$), inequality ($\S6.5$), binary logical operators ($\S6.3$) and logical not ($\S??$).

```
BoolType ::= bool
```

The following illustrates a simple example of the bool type:

```
// Determine whether item is contained in list or not
function contains([int] list, int item) => bool:
    // examine every element of list
    for 1 in list:
        if 1 == item:
            return true
    // done
    return false
```

This function determines whether or not a given integer value is contained within a list of integers. If so, it returns true, otherwise it returns false.

4.2.3 Bytes

The type **byte** represents the set of eight-bit sequences, whose values are expressed numerically using 0 and 1 followed by b (e.g. 00101b). The set of values defined by the **byte** type is the set of all 256 possible combinations of eight-bit sequences. Variables of **byte** type support equality ($\S6.5$), inequality ($\S6.5$), bitwise operators ($\S6.4$), bitwise complement (\S ??) and shift operators ($\S6.9$).

```
ByteType ::= byte
```

The following illustrates a simple example of the byte type:

```
// convert a byte into a string
function toString (byte b) => string:
    string r = "b"
    for i in 0..8:
        if (b & 00000001b) == 00000001b:
            r = "1" ++ r
        else:
            r = "0" ++ r
        b = b >> 1
    return r
```

This illustrates the conversion from a **byte** into a **string**. The conversion is performed one digit at a time, starting from the rightmost bit.

Notes. Unlike for many languages, there is no representation associated with a byte. For example, to extract an integer value from a byte, it must be explicitly decoded according to some representation (e.g. two's compliment) using an auxillary function (e.g. Byte.toInt()).

4.2.4 Integers

The type int represents the set of arbitrary-sized integers, whose values are expressed as a sequence of one or more numerical or hexadecimal digits (e.g. 123456, 0xffaf, etc). Variables of int type support equality (§6.5), inequality (§6.5), comparators (§6.5), addition (§6.10), subtraction (§6.10), multiplication (§6.10), division (§6.10), remainder (§6.10) and negation (§??) operations.

```
IntType ::= int
```

The following illustrates a simple example of the int type:

```
function fib(int x) => int:
   if x <= 1:
       return x
   else:
      return fib(x-1) + fib(x-2)</pre>
```

This illustrates the well-known recursive method for computing numbers in the *fibonacci* sequence.

Notes. Since integers in Whiley are of arbitrary size, *integer overflow* is not possible. This contrasts with other languages (e.g. Java) that used *fixed-width* number representations (e.g. 32bit two's complement). Furthermore, there is nothing equivalent to the constants found in such languages for representing the uppermost and least integers expressible (e.g. Integer.MIN_VALUE and Integer.MAX_VALUE, as found in Java).

4.2.5 Rationals

The type real represents the set of arbitrary-sized rationals, whose values are expressed as a sequence of one or more numerical digits separated by a period (e.g. 1.0, 0.223, 12.55, etc). The

set of values defined by the type **real** is the (infinite) set of all integer pairs, where the first element is designated the numerator, and the second designated the denominator. Variables of **real** type support equality ($\S6.5$), inequality ($\S6.5$), comparators ($\S6.5$), addition ($\S6.10$), subtraction ($\S6.10$), multiplication ($\S6.10$), division ($\S6.10$), remainder ($\S6.10$) and negation ($\S2.10$) operations. Variables of type **real** also support the *rational destructuring assignment* to extract the numerator and denominator (illustrated below).

```
RealType ::= real
```

The following illustrates a simple example of the real type:

This illustrates the well-known function for computing the *floor* of a **real** variable x (i.e. the greatest integer not larger than x). The rational destructuring assignment is used to extract the numerator and denominator of the parameter x.

4.2.6 Characters

The type **char** represents the set of unicode characters, whose values are expressed as an arbitrary character between quotes (e.g. 'c', '0', '\$', etc). Variables of **char** type support equality (§6.5), inequality (§6.5), comparators (§6.5), addition (§6.10), subtraction (§6.10), multiplication (§6.10), division (§6.10), remainder (§6.10) and negation (§??) operations.

```
CharType ::= char
```

The following illustrates a simple example of the **char** type:

```
function isUpperCase(char c) => bool:
    return 'A' <= c && c <= 'Z'</pre>
```

This illustrates a very simple function for determining whether an ASCII character is uppercase or not.

4.2.7 Any

The type **any** represents the type whose variables may hold any possible value. Thus, **any** is the *top type* (i.e. \top) in the lattice of types and, hence, is the supertype of all other types. Variables of **any** type support only equality (§6.5), inequality comparisons (§6.5) and *runtime type tests*. Finally, unlike the majority of other types, there are no *values* of type **any**.

```
AnyType ::= any
```

The following illustrates a simple example of the any type:

```
function toInt(any val) => int:
   if val is int:
      return val
   else if val is real:
      return Math.floor(val)
   else:
      return 0 // default value
```

Here, the function toInt accepts any valid Whiley value, which includes all values of type int, real, collections, records, etc. The function then inspects the value that it has been passed and, in the case of values of type int and real, returns an integer approximation; for all other values, it returns 0.

Notes. The any type is roughly comparable to the Object type found in pure object-oriented languages. However, in impure object-oriented languages which support primitive types, such as Java, this comparison often falls short because Object is not a supertype of primitives such as **int** or long.

4.2.8 Void

The **void** type represents the type whose variables cannot exist (i.e. because they cannot hold any possible value). Thus, **void** is the *bottom type* (i.e. \perp) in the lattice of types and, hence, is the *subtype* of all other types. Void is used to represent the return type of a method which does not return anything. Furthermore, it is also used to represent the element type of an empty list of set. Finally, unlike the majority of other types, there are no *values* of type **void**.

```
VoidType ::= void
```

The following example illustrates several uses of the void type:

```
// Attempt to update first element
method update1st(&[int] list, int value) => void:
    // First, check whether list is empty or not
    if *list != [void]:
        // Then, update 1st element
        (*list)[0] = x
        // done
```

Here, the method update1st is declared to return **void** — meaning it does not return a value. Instead, this method updates some existing state accessible through the reference list. Within the method body, the value accessible via this reference is compared against the [**void**] (i.e. the *empty list*).

4.3 Tuples

A tuple type describes a compound type made up of two or more elements in sequence, whose values are expressed as sequences of values separated by a comma (e.g. 1, 2, 2.0, 3.32, 3.45, etc). Tuples are similar to records, except that fields are effectively anonymous. Variables of tuple type support equality ($\S6.5$) and inequality ($\S6.5$) operations, as well the *tuple destructuring assignment* to extract elements (illustrated below).

```
TupleType ::= ( Type(, Type)+)

TuplePattern ::= ( TypePattern(, TypePattern)+) [ Ident ]
```

The following example illustrates several uses of tuples:

```
function swap(int x, int y) => (int,int):
    return y,x
```

This function accepts two integer parameters, and returns a tuple type containing two integers. The function simple reverses the order that the values occur in the tuple passed as a parameter.

4.4 Records

A record type describes a compound made of from one or more *fields*, each of which has a unique name and a corresponding type. Variables of record type support equality ($\S6.5$), inequality ($\S6.5$) and field access ($\S6.11$) operations, as well as field assignment (\S ??).

```
RecordType ::= { MixedType(, MixedType)*[, ...]}
```

Records use *mixed types* for defining fields (see §??), meaning that field names may be mixed within their type. This is primarily useful for fields of function or method type (see below). Records using the \ldots notation are referred to as *open records* (e.g. {int x, ...}), otherwise they are referred to as *closed records* (e.g. {int x, int y}). Open records represent all records containing *at least* the given fields, whilst closed records represent those containing *exactly* the given fields. The following example illustrates an open record type:

```
type Writer is {
    method write([byte]) => int,
    ...
}

type PrintWriter is {
    method write([byte]) => int,
    method println(string) => void,
    ...
}

// Create print writer if not already one
PrintWriter create(Writer writer):
    if writer is PrintWriter:
        return writer
    else:
        return new PrinterWriter(writer)
```

The above illustrates two open records Writer and PrintWriter. The former has one field (write), whilst the latter has two fields (write and println). The above also illustrates use of mixed types. For example, the field "write" is declared as "method write([byte]) => int" which mixes together the field name (i.e. "write") with its type (i.e. "method([byte]) => int").

4.5 References

Reference types in Whiley represent references to variables, such as those allocated in the heap. They are similar to references or pointers found in many imperative and object-oriented languages (e.g. C/C++, Java, C#, etc). A type &T represents a reference to a variable of type T. Variables of reference type support equality ($\S6.5$), inequality ($\S6.5$) and dereference (\S ??) operations, as well as dereference assignment (\S ??).

```
ReferenceType ::= & Type
```

The following example illustrates reference types:

```
// Swap contents of heap-allocated int variables
method swap (&int pX, &int pY):
    int tmp = *pX
    *pX = *pY
    *pY = tmp
```

The above illustrates a method which accepts two references to variables of type int that may refer to the same variable. The method simply swaps the contents of the variables to which they refer.

4.6 Nominals

Nominal types represent user-defined types declared within one or more Whiley source files. Nominal types provide a mechanism for enforcing *information hiding*, and also for constructing *recursive types* (§4.12). All nominal types have an underlying — or, *concrete* — type which may or may not be visible within a given Whiley source file. Nominal types with *visible* underlying types are indistinguishable from their underlying type, and support all operations therein. Nominal types with *invisible* underlying types support only equality (§6.5) and inequality (§6.5) operations. Furthermore, to enforce information hiding, nominal types cannot be destructured using runtime type tests.

```
NominalType ::= Ident
```

The following example illustrates nominal types:

```
// Using a nominal type to construct a recursive type
public type LinkedList is null | { int data, LinkedList next }

// Using a nominal type to enforce information hiding
protected type Hidden is { int x, int y }
```

Three different uses for nominal types are illustrated here. The type LinkedList is declared **public**, meaning that: firstly, it can be referred to by name from any other source file; secondly, its underlying type is visible to any other source file. Within the declaration of LinkedList a reference to itself is used to define a recursive type (§4.12). Finally, the type hidden has the concrete type "{int x, int y}" and is declared **protected**, meaning that: firstly, it can be referred to by name from any other source file; secondly, that its underlying type is invisible to all other source files.

4.7 Collections

Collection types in Whiley describe compound values constructed from arbitrarily many values.

```
CollectionType ::= SetType
| MapType
| ListType
```

4.7.1 Sets

A set type describes set values whose elements are subtypes of the element type. For example, $\{1,2,3\}$ is an instance of set type $\{int\}$; however, $\{1.345\}$ is not. Variables of set type support equality (§6.5), inequality (§6.5), union (§??), intersection (§??), difference (§??) and element-of (§??) operations.

```
SetType ::= { Type }
```

The following example illustrates set types:

```
// Adjacency list representation
type Graph is ([{uint}] es)

function depthFirstSearch(uint v, Graph graph, {int} visited) => {int}

//
visited = visited + {v}
// Traverse edges not yet visited
for w in graph[v]:
    if ! (w in visited):
        visited = depthFirstSearch(w, graph, visited)

// Done
return visited
```

The above illustrates a simple implementation of the well-known *depth-first search* algorithm. In the example, a Graph is a list of sets of edge targets, where any w in g[v] describes an edge from v to w in the graph. The visited set is used to maintain a list of previously seen vertices, in order to prevent the same vertex from being visited more than once.

4.7.2 Maps

A map represents a one-many mapping from variables of one type to variables of another type. For example, the map type $\{int=>real\}$ represents a map from integers to real values. A valid instance of this type might be $\{1=>1.2,2=>3.0\}$. Variables of map type support equality ($\S6.5$), inequality ($\S6.5$), access ($\S6.11$), union ($\S??$), intersection ($\S??$), difference ($\S??$) and element-of ($\S??$) operations.

```
MapType ::= { Type => Type }
```

The following example illustrates map types:

```
type Expr is int | string // simple expression forms

int evaluate(Expr e, {string=>int} environment)
   if e is int:
        // expression is constant, so return this directly
        return e
   else:
        // expression is variable, so look up its value in environment
        return environment[e]
```

The above illustrates a function for evaluating simple expressions which are either integer constants or variable names. To evaluate an expression which is an integer constant, we simply return that constant. To evaluate an expression which is a variable name, we look up the current value of that variable in an environment which maps variable names to integer constants.

4.7.3 Lists

A list type describes list values whose elements are subtypes of the element type. For example, [1,2,3] is an instance of list type [int]; however, [1.345] is not. Variables of list type support equality (§6.5), inequality (§6.5), append (§6.7), sublist (§??) and element-of (§??) operations.

```
ListType ::= [ Type ]
```

The following example illustrates map types:

```
function add([int] v1, [int] v2) => ([int] v3)

//
  int i=0
  while i < |v1|:
     v1[i] = v1[i] + v2[i]
     i = i + 1

//
  return v1</pre>
```

The above illustrates a simple function which adds two integer lists together. The function's precondition requires that both input list have the same length, whilst its postconditions ensures that this matches the length of the output.

4.8 Functions and Methods

A function or method type describes the signature of a function or method. These types enable functions or methods to be passed around as values in Whiley and are often referred to as *functors*. This enables a degree of polymorphism in the language, where the exact function or method to be called is unknown. Variables of function or method type support equality ($\S 6.5$) and inequality ($\S 6.5$).

```
FunctionType ::= function ( [Type (, Type)*]) => Type

MethodType ::= method ( [Type (, Type)*]) => Type
```

The following example illustrates function types:

```
type Fun is function(int) => int

function map([int] items, Fun fn) => [int]:
    //
    for i,v in items:
        items[i] = fn(v)
    //
    return items
```

The above illustrates the well-known *map* function, which maps all elements of a list according to a given function.

4.9 Unions

A union type is constructed from two or more component types and contains any value held in any of its components. For example, the type **null**|**int** is a union which holds either an integer value or **null**. The set of values defined by a union type T1|T2 is exactly the union of the sets defined by T1 and T2. In general, variables of union type support only equality (§6.5), inequality comparisons (§6.5) and *runtime type tests* (see §4.13 for exceptions to this).

```
UnionType ::= IntersectionType ( | IntersectionType )*
```

The following example illustrates a union type:

Here, a union type is used to construct a more expressive return value. If no matching element is found, **null** is returned (rather than e.g. -1).

4.10 Intersections

An intersection type is constructed from two or more component types and contains any value held in all of its components. For example, the type [int] & [bool] is an intersection which hold any value which is both an instance of [int] and [bool] (in fact, only the empty list meets this criteria). Intersections are used to type variables on the true branch of a runtime type test. The set of values defined by an intersection type T1&T2 is exactly the intersection of the sets defined by T1 and T2. In general, variables of intersection type support only equality (§6.5), inequality comparisons (§6.5) and runtime type tests (see §4.13 for exceptions to this).

```
IntersectionType ::= TermType(& TermType)*
```

The following example illustrates an intersection type:

```
type Reader is {
  method read(int) => [byte],
  ...
}
type Writer is {
  method write([byte]) => int,
  ...
}
type ReaderWriter is Reader & Writer
```

Here, the type Reader is defined as any record containing a read(int) method, whilst the type Writer is defined as any record containing a write([byte]) method. Then, the intersection type ReaderWriter is defined as any record containing both a read(int) and write([byte]) method.

4.11 Negations

A negation type is constructed a component type and contains any value *not* held in its component. For example, the type !int is a negation which holds any non-integer value. Negations are used to type variables on the false branch of a runtime type test. The set of values defined by a negation type ! T_1 is exactly the set of all values less those defined by T_1 . In general, variables of negation type support only equality (§6.5), inequality comparisons (§6.5) and *runtime type tests* (see §4.13 for exceptions to this).

```
NegationType ::= [! TermType
```

The following example illustrates a negation type:

```
function f(any item) => !null:
   if item is null:
      return 0
   else:
      return item
```

Here, the function f() accepts a parameter of any type, and returns a value which is permitted to be anything except **null**. The above also illustrates how the type test operator (§??) retypes variables on the false branch using negation types.

4.12 Recursive Types

Recursive types describe tree-like structures of arbitrary depth. For example, linked lists, binary trees, quad trees, etc can all be described using recursive types. Recursive types have no explicit syntax and, instead, are declared indirectly in terms of themselves using one or more nominal types (§4.6).

The following example illustrates a simple recursive type:

```
type Node is { Tree left, Tree right, int data }
type Tree is null | Node
function sizeOf(Tree t) => int:
```

```
if t == null:
    return 0
else:
    return 1 + sizeOf(t.lhs) + sizeOf(t.rhs)
```

Here, the type Tree is recursive because it is defined in terms of itself. An instance of type Tree is a sequence of nested records which is arbitrarily deep, and whose branches are terminated by null. The function sizeOf() traverses an arbitrary instance of Tree and returns the number of Nodes it contains.

4.13 Effective Types

An effective type is a union of types which all contain some property (e.g. a union of lists). This common property allows the effective type to support more operations than possible for an arbitrary union (§4.9).

4.13.1 Effective Tuples

An effective tuple is a union of tuple types. For example, $(int,int) \mid (real,real)$ is an effective tuple. An effective tuple type supports all operations valid for a tuple type (§4.3).

4.13.2 Effective Records

An effective record is a union of record types. For example, {int f, int g}|{real f, int h} is an effective record. An effective record provides access to fields common to all records in the union. For example, the type {int f, int g}|{real f, int h} can be viewed as having an effective type of {int|real f, ...} and, hence, read access to field f is given.

4.13.3 Effective Collections

An effective collection is a union of collection types. For example, [int] | [real] is an effective list. An effective collection supports all operations valid for a collection type (§4.7). For example, the type [int] | [real] can be viewed as having an effective type of [int|real] and, hence, read access to its length and elements is given.

4.14 Semantics

Although types are abstract entities we can (for the most part) imagine them as describing sets of *abstract values*. For example, int|null denotes the set of values containing exactly the (infinite) set of integers and null (i.e. $\mathbb{Z} \cup \{null\}$). This is often referred to as a set-theoretic interpretation of types [15;16;17;18]. Under this interpretation, for example, one type *subtypes* another if the set of values it denotes is a *subset* of the other (see § 4.14.2 for more).

We specify the meaning of types by formalising a set theoretic interpretation of them over the language of values given in Figure 4.1. To minimise confusion, care is taken in the figure to ensure that abstract values are represented canonically. For example, "2 / 4" is not a valid abstract value since "1 / 2" is its canonical representation. Likewise, " $\{2,1\}$ " is not a valid abstract value, with " $\{1,2\}$ " being its canonical representation. Figure 4.1 separates abstract values into distinct categories (e.g. integers, rationals, tuples, etc). These distinctions are significant. For example, "0" is distinct from "0 / 0". Similarly, byte values are not expressed using the digits 0 and 1 (as might be expected), but in terms of the characters t and f. This ensures binary values are distinct from integer values.

How zero

An evaluation function [T] is defined which returns the set of values associated with a type T. For example, $[bool] = \{true, false\}$, $[int] = \mathbb{Z}$, etc. This function is defined as follows:

```
(null value)
::= null
             true | false
                                                                                                                                    (boolean values)
                                                             if b \in \{t, f\}^8
                                                                                                                                          (byte values)
                                                             if \mathbf{i} \in \mathbb{Z}
             i
                                                                                                                                      (integer values)
                                                             if \mathtt{i} \in \mathbb{Z}, \mathtt{n} \in \mathbb{N} and \text{gcd}(\mathtt{i}_1,\mathtt{i}_2) = 1
             i/n
                                                                                                                                    (rational values)
                                                             if \mathtt{n} \in \mathbb{N}
                                                                                                                                  (character values)
             (\mathtt{v_1}, \ldots, \mathtt{v_n})
                                                                                                                                         (tuple values)
                                                            if \forall i.v_i < v_{i+1}
                                                                                                                                             (set values)
             \{\mathtt{v_1} \Rightarrow \mathtt{v_1'}, \ldots, \mathtt{v_n} \Rightarrow \mathtt{v_n'}\} \quad \text{if } \forall \mathtt{i.v_i} < \mathtt{v_{i+1}}
                                                                                                                                          (map values)
                                                                                                                                            (list values)
                                                                                                                                              (locations)
```

Figure 4.1: The language of abstract values used to formalise the meaning of types in Whiley, where \mathbb{Z} is the (infinite) set of integers, \mathbb{N} the (infinite) set of naturals and gcd() returns the Greatest Common Divisor of two values (e.g. using Euclid's well-known algorithm).

Definition 1 (Type Semantics) Every type descriptor T is characterized by the set of values it accepts, given by [T] and defined as follows:

Here, the *domain* of all possible values is given by \mathbb{D} , whilst the set of all integers is given by \mathbb{Z} . Furthermore, if T is a type descriptor, then [T] is its underlying type.

4.14.1 Equivalences

Since types are defined in terms of the set of values they represent, it is perfectly possible for two distinct type descriptors to describe the same underlying type. For example, <code>int|null</code> is considered equivalent to <code>null|int</code>. Whilst this case is fairly easy to spot, there are some cases which are not so obvious. The definition of equivalence is given as follows:

Definition 2 (Type Equivalence) Two type descriptors T_1 and T_2 are said to be equivalent, denoted by $T_1 \equiv T_2$, iff $[\![T_1]\!] = [\![T_2]\!]$.

Based on the above definition, we identify a number of such equivalences to illustrate:

- ! any is equivalent to void and, conversely, any is equivalent to !void
- int&!int is equivalent to void and, conversely, int |!int is equivalent to any
- {int | null f} is equivalent to {int f} | {null f}
- {int | null f} & {bool | null f} is equivalent to {null f}

Under Definition 2, an infinite number of equivalences exist between the type descriptors of Whiley, and we cannot list them all here.

4.14.2 Subtyping

Types in Whiley support the notion of *subtyping* where one type may be a *subtype* for another. For example, the type **int** is a subtype of **any**. Likewise, **bool** is a subtype of **bool** | **null**. The *subtyping operator* is denoted by " \leq "; for example, $T_1 \leq T_2$ indicates that type T_1 is a subtype of T_2 . The subtyping operator is *reflexive*, *transitive* and *anti-symmetric* with respect to the underlying types involved.

The subtyping operator is regarded as an algorithm for determining whether the type described by one type descriptor is a subtype of another. The implementation of this algorithm is not straightforward and a full discussion of it is beyond the scope of this document. Indeed, there are many possible implementations of this operator. Nevertheless, there any valid implementation of this operator must exhibit two desirable properties:

Definition 3 (Subtype Soundness) A subtype operator, \leq , is sound if, for any types T_1 and T_2 , it holds that $T_1 \leq T_2 \Longrightarrow [T_1] \subseteq [T_2]$.

Definition 4 (Subtype Completeness) A subtype operator, \leq , is complete if, for any types T_1 and T_2 , it holds that $[\![T_1]\!] \subseteq [\![T_2]\!] \Longrightarrow T_1 \leq T_2$.

A subtype operator which exhibits both of these properties is said to be sound and complete.

Chapter 5

Statements

The execution of a Whiley program is controlled by *statements*, which cause effects on the environment. However, statements in Whiley do not produce values. *Compound statement* statements may contain other statements.

5.1 Blocks

A statement block is a sequence of zero or more consecutive statements which have the same indentation. Statement blocks are used to group statements together when constructing compound statements. For example:

```
function sum([int] items) => int:
    // outer block begins
    int r = 0
    int i = 0
    while i < |items|:
        // inner block begins
        r = r + items[i]
        i = i + 1
        // inner block ends
//
return r
// outer block ends</pre>
```

The above example contains two statement blocks, one nested inside the other. The outer block demarcates the body of the sum () function, whilst the inner block demarcates the body of the while statement.

5.2 Assert Statement

Represents an assert statement of the form "assert e", where e is a boolean expression. A fault will be raised at runtime if the asserted expression evaluates to false; otherwise, execution will proceed normally. At verification time, the verifier is forced to ensure that the asserted expression is true for all possible execution paths. This allows the programmer to specify and check something he/she believes to be true at a given point in the program.

```
AssertStmt ::= assert Expr
```

The following illustrates an assert statement:

```
function abs(int x) => int:
    if x < 0:
        x = -x
    assert x >= 0
    return x
```

Here, an assertion is used to check that the value being returned by the abs () is non-negative. Since this is a true statement of the function, this statement will never raise a fault.

5.3 Assignment Statement

As assignment statement is of the form leftHandSide = rightHandSide. Here, the rightHandSide is any expression, whilst the leftHandSide must be an LVal — that is, an expression permitted on the left-hand side of an assignment. At runtime, the value generated by evaluating the right-hand side must be a subtype (§4.14.2) of the left-hand side.

```
AssignStmt ::= LVal = Expr
```

The following illustrates different possible assignment statements:

```
x = y  // variable assignment
x.f = y  // field assignment
x[i] = y  // list assignment
x[i].f = y  // compound assignment
```

The last assignment here illustrates that the left-hand side of an assignment can be arbitrarily complex, involving nested assignments into lists and records.

5.4 Assume Statement

An assume statement is of the form "assume e", where e is a boolean expression. A fault will be raised at runtime if the assumed expression evaluates to false; otherwise, execution will proceed normally. At verification time, the verifier will automatically assume that the given expression holds. Thus, assume statements provide a way for the programmer to override the verifier. This is useful where the verifier is unable to establish something that the programmer knows to be true. Care must be taken to ensure that the assumed expression really does hold.

```
AssumeStmt ::= assume Expr
```

The following illustrates an assume statement:

```
function abs(int x) => (int y)
ensures y >= 0:
    //
    assume x >= 0
    return x
```

Here, the programmer has used an assumption to ensure this function passes verification. This would not appear to be safe in this case, and may lead to a fault at runtime.

- 5.5 Break Statement
- **5.6** Continue Statement
- **5.7** Debug Statement
- 5.8 Do/While Statement

```
DoWhileStmt^\ell ::= do : Block^\gamma while Expr (where Expr)^* (where \ell < \gamma)
```

Description.

Notes.

5.9 For Statement

```
For Stmt \ell ::= for Var Pattern in Expr ( where Expr )* : Block \gamma (where \ell < \gamma)
```

Description.

Notes.

5.10 If Statement

An **if** statement conditionally executes a statement block based on the outcome of one or more expressions. Chaining of **if** statements is permitted, and an optional **else** branch may be given. The expression(s) are referred to as *conditions* and must be boolean expressions. The first block is referred to as the *true branch*, whilst the optional **else** block is referred to as the *false branch*.

```
\text{IfStmt}^{\ell} \ ::= \ \boxed{\text{if Expr}: Block}^{\gamma} \left( \boxed{\text{else}} \ \boxed{\text{if Expr}: Block}^{\omega_i} \right)^* \\ \left[ \boxed{\text{else}: Block}^{\phi} \right]  \left( \text{where } \ell < \gamma \text{ and } \forall i.\ell < \omega_i \text{ and } \ell < \phi \right)
```

The following illustrates an if statement:

```
function max(int x, int y) => int:
  if(x > y):
    return x
```

```
else if(x == y):
    return 0
else:
    return y
```

Here, we see an **if** statement with two conditional outcomes and one default outcome.

5.11 While Statement

A while statement repeatedly executes a statement block until an expression (the condition) evaluates to false. Optional **where** clause(s) are permitted which, together, are commonly referred to as the loop invariant.

The following illustrates an while statement:

```
function sum([int] xs) => int:
  int r = 0
  int i = 0
  while i < |xs| where i >= 0:
    r = r + xs[i]
    i = i + 1
  return r
```

Here, we see a simple **while** statement which sums the elements of variable xs, storing the result in variable r. A loop invariant is given which establishes that variable i is non-negative.

Notes. When multiple **where** clauses are given, these are combined using a conjunction to form the loop invariant. The combined invariant must hold on entry to the loop and after each iteration. Thus, when the condition evaluates to false, the loop invariant is guaranteed to hold. However, the loop invariant need not hold when the loop is exited using a break ($\S 5.5$) statement.

5.12 Return Statement

A *return statement* has an optional expression referred to as the *return value*. At runtime, this statement returns control to the caller of the enclosing function or method. At verification time, the verifier will ensure the returned value meets the postcondition of the enclosing function or method.

```
ReturnStmt ::= return [Expr]
```

The following illustrates a return statement:

```
function f(int x) => int:
    return x + 1
```

Here, we see a simple simple function which returns the increment of its parameter x using a return statement.

Notes. The returned expression (if there is one) must begin on the same line as the statement itself.

5.13 Skip Statement

5.14 Switch Statement

```
SwitchStmt ::=
```

Description.

Notes.

5.15 Throw Statement

A *throw statement* causes an exception to thrown which, if not caught locally, causes an *abrupt termination* of the current function or method. Functions or methods which may terminate abruptly must declare appropriate an throws clause (§3.5.4,§3.5.5) which contains all potentially thrown exceptions.

```
ThrowStmt ::= throw Expr
```

The following illustrates a throw statement:

```
function parseInt(int pos, string input) => (int,int)
// Throws a syntax error if the string is malformed
throws SyntaxError:
    //
    int start = pos
    // check for negative input
    if pos < |input| && input[pos] == '-':
        pos = pos + 1
    // match remainder
    while pos < |input| && Char.isDigit(input[pos]):
        pos = pos + 1
    // check for error
    if pos == start:
        throw SyntaxError("Missing_number", start, pos)
    // done
    return Int.parse(input[start..pos]), pos</pre>
```

Here, we see a simple function which parses a string into an integer. The function declares that a SyntaxError may be thrown. This is required for two reasons: firstly, the input contains no digits then an SyntaxError is thrown by this function. Additionally, the function Int.parse() is declared to throw a SyntaxError and, since it is not caught, this declaration must be propagated.

5.16 Try/Catch Statement

```
TryCatchStmt ::=
```

Description.

Notes.

5.17 Variable Declaration Statement

A variable declaration statement has an optional expression assignment referred to as a variable initialiser. If an initialiser is given, this will be evaluated and assigned to the variable when the declaration is executed.

```
VarDecl ::= Type Ident [ = Expr]
```

Some example variable declarations are:

```
int x
int y = 1
int z = x + y
```

Here we see three variable declarations. The first has no initialiser, whilst the second and third have initialisers.

Chapter 6

Expressions

Expression blah blah.

6.1 Tuple Expressions

```
TupleExpr ::= UnitExpr (, UnitExpr)+
```

Description.

Examples.

Notes.

6.2 Unit Expressions

```
UnitExpr ::= LogicalExpr
```

Description.

Examples.

Notes.

6.3 Logical Expressions

```
LogicalExpr ::= LogicalOrExpr ==> LogicalExpr

LogicalOrExpr ::= LogicalAndExpr

LogicalOrExpr || LogicalAndExpr

LogicalAndExpr ::= BitwiseExpr

LogicalAndExpr && BitwiseExpr
```

Description.

Examples.

Notes.

6.4 Bitwise Expressions

Description.

Examples.

Notes.

6.5 Condition Expressions

```
ConditionExpr ::=
```

Description.

Examples.

Notes.

6.6 Quantifier Expressions

```
QuantExpr ::= ( no | some | all ) {

Ident in Expr (, Ident in Expr)+ | LogicalExpr
}
```

Description.

Examples.

Notes.

6.7 Append Expressions

```
AppendExpr ::= RangeExpr ( ++ RangeExpr)*
```

Description.

Examples.

Notes.

6.8 Range Expressions

```
RangeExpr ::= ShiftExpr [ .. ShiftExpr ]
```

Description.

Examples.

Notes.

6.9 Shift Expressions

```
ShiftExpr ::= AdditiveExpr[( << | >> ) AdditiveExpr]
```

Description.

Examples.

Notes.

6.10 Additive/Multiplicative Expressions

```
AdditiveExpr ::=
MultiplicativeExpr ::=
```

Description.

Examples.

Notes.

6.11 Access Expressions

```
AccessExpr ::=
```

Description.

Examples.

Notes.

6.12 Term Expressions

```
TermExpr ::=
```

Description.

Examples.

Notes.

6.13 Dereference Expressions

The dereference operation " $e \rightarrow f$ " is a short-hand notation for "($\star e$) . f" and can be used when e has effective record type ($\S 4.13.2$).

Glossary

- **abrupt termination** A statement terminates abruptly if a subexpression causes an exception to be thrown. This includes exceptions thrown (and not caught) by an invoked function or method. 34
- **block comment** A block comment begins with "/*" and continues until the end-of-comment marker "*/". 7
- boolean expression An expression which evaluates to a value of type bool. 12, 30–32, 40
- **compilation group** A group of one or more source files being compiled together. 10
- **compilation unit** A single unit of compilation. In Whiley, this includes source files and also binary WyIL files. 10, 11
- **compound statement** A statement (e.g. **if**, **while**, etc) which may contain blocks of other statements. 30
- **constant declaration** A source-level declaration which associates a name with a constant expression. The full name of the declared entity is determined from the package and name of the enclosing source file.. 10
- **default package** The top-level package which has no name, and is considered to be a "global" package.. 10
- **expression** A combination of constants, variables and operators that, when evaluated, produce a single value. Expressions in certain circumstances may have side effects. 36, 40, 41
- **fault** A fault is raised when an unrecoverable error in the program occurs. For a verified program, no faults are possible except to indicate an out-of-memory failure.. 30, 31
- **function declaration** A source-level declaration which defines a named function. The full name of the declared entity is determined from the package and name of the enclosing source file.. 10
- **indentation syntax** A lexical organisation of source files where indentation is significant and is used to group statements and blocks. 6
- intersection type A type formed by combining two or more types together (e.g. [int] & [any]), such that it includes any value contained in both. 15
- **line comment** A line comment begins with "//" and continues until the end of line. 7
- literal A source-level entity which describes a value of primitive type. 8
- loop invariant A boolean expression which must hold on every iteration of a loop. 13, 14, 33

- **method declaration** A source-level declaration which defines a named method. The full name of the declared entity is determined from the package and name of the enclosing source file.. 10
- name resolution The process of determining the fully qualified name of an identifier within a source file. Names are first resolved within the same source file, and then by searching the list of imported entities in reverse order. 11
- **negation type** A type formed from another (e.g. !int), such that it includes any value not contained in the other. 15
- package A unit of hierarchical organisation within the Whiley namespace.. 10
- **postcondition** A logical condition over the parameters and returns of a function or method which must be true immediately after execution of that function or method.. 13, 14, 24, 33
- **precondition** A logical condition over the parameters of a function or method which must be true immediately prior to execution of that function or method.. 13, 14, 24
- **safety critical system** A system which operates in a high-risk setting where failure can lead to loss of life, injury, significant damage or environmental harm. 4
- **source file** A file in which source code is located. Source files for the Whiley programming language have the extension .whiley. In Whiley, source files must be compiled into a binary form before they can be executed.. 6, 10, 12–14, 22, 40, 41
- **statement** An program instruction which has an effect on the environment when executed, but does not produce a value. 30, 41
- **statement block** A sequence of zero or more consecutive statements with the same indentation. 6, 30, 32
- **type** An abstract entity which represents the set of values a given variable may hold, or a given expression may evaluate to.. 15, 28, 40, 41
- **type declaration** A source-level declaration which associates a name with a type descriptor. The full name of the declared entity is determined from the package and name of the enclosing source file.. 10
- **type descriptor** A source-level description of an underlying type. Unlike many languages, type descriptors and types are quite distinct in Whiley as, for example, two distinct descriptors may describe the same underlying type. 15, 28, 41
- **type pattern** A source-level description of an underlying type (similar to a type descriptor) where one or more variables are associated with its subcomponent(s).. 16
- union type A type formed by combining two or more types together (e.g. int | null), such that it includes any value contained in either. 15
- variable declaration A statement which declares one or more variable(s) for use in a given scope. Each variable is given a type which limits the possible values it may hold, and may not already be declared in an enclosing scope. 35, 41
- variable initialiser An optional expression used to initialise variable(s) declared as part of a variable declaration. 35
- **verifying compiler** A compilers which employs automated mathematical and logical reasoning to check the correctness of the programs that it compiles. 4
- WyIL file A compiled (i.e. binary) form of a Whiley source file. 10, 40

Bibliography

- [1] David J. Pearce. Getting Started with Whiley. 2014.
- [2] Nancy G. Leveson and Clark S. Turner. An investigation of the Therac-25 accidents. *IEEE Computer*, 26(7):18–41, 1993.
- [3] Mark W. Eichin and Jon A. Rochlis. With microscope and tweezers: An analysis of the internet virus of November 1988. In *Proc. IEEE Symposium on Research in Security and Privacy*, pages 326–343, 1989.
- [4] Software problem led to system failure at dhahran, saudi arabia, gao report #b-247094, 1992.
- [5] Ariane 5: Flight 501 failure. report by the enquiry board. Technical report, European Space Agency, 1996.
- [6] Tony Hoare. The verifying compiler: A grand challenge for computing research. *Journal of the ACM*, 50(1):63–69, 2003.
- [7] S. King. A Program Verifier. PhD thesis, Carnegie-Mellon University, 1969.
- [8] L. Peter Deutsch. An interactive program verifier. Ph.d., 1973.
- [9] D. I. Good. Mechanical proofs about computer programs. In *Mathematical logic and program-ming languages*, pages 55–75, 1985.
- [10] D. C. Luckham, S. M. German, F. W. von Henke, R. A. Karp, P. W. Milne, D. C. Oppen, W. Polak, and W. L. Scherlis. Stanford pascal verifier user manual. Technical Report CS-TR-79-731, Stanford University, Department of Computer Science, 1979.
- [11] David L. Detlefs, K. Rustan M. Leino, Greg Nelson, and James B. Saxe. Extended static checking. SRC Research Report 159, Compaq Systems Research Center, 1998.
- [12] Cormac Flanagan, K. Rustan M. Leino, Mark Lillibridge, Greg Nelson, James B. Saxe, and Raymie Stata. Extended static checking for Java. In *Proc. PLDI*, pages 234–245, 2002.
- [13] G. T. Leavens, Y. Cheon, C. Clifton, C. Ruby, and D. R. Cok. How the design of JML accommodates both runtime assertion checking and formal verification. *Science of Computer Programming*, 55(1-3):185–208, March 2005.
- [14] Mike Barnett, K. Rustan, M. Leino, and Wolfram Schulte. The spec# programming system: An overview. Technical report, Microsoft Research, 2004.
- [15] Alexander Aiken and Edward L. Wimmers. Type inclusion constraints and type inference. In *Proceedings of the ACM conference on Functional Programming Languages and Computer Architecture (FPCA)*, pages 31–41. ACM Press, 1993.
- [16] Flemming M. Damm. Subtyping with union types, intersection types and recursive types. volume 789 of *LNCS*, pages 687–706. 1994.

- [17] Castagna and Frisch. A gentle introduction to semantic subtyping. In *Proc. ICALP*, pages 198–199, 2005.
- [18] A. Frisch, G. Castagna, and V. Benzaken. Semantic subtyping: Dealing set-theoretically with function, union, intersection, and negation types. *JACM*, 55(4):19:1–19:64, 2008.